Curso: 2006-2007

INGENIERÍA TÉCNICA INDUSTRIAL - Esp. Mecánica

FUNDAMENTOS MATEMÁTICOS DE LA INGENIERÍA

Práctica nº 6: Cálculo diferencial de funciones de una variable

RESUMEN: Estudiamos en esta práctica el tratamiento con MAPLE de los conceptos básicos del cálculo diferencial de funciones de una variable y sus aplicaciones: estudio de máximos y mínimos y polinomio de Taylor.

COMANDOS MÁS IMPORTANTES:

D(f)(x), (D@@n)(f)(x), unapply(), maximize()

Derivada de una función

La derivada de una función f*(x)* en un punto *a* viene dada por

$$f'(a) = \lim_{h \to 0} \frac{f(a+h) - f(a)}{h}$$

Geométricamente f '(*a*) es la pendiente de la recta tangente a la curva y=f(x) en el punto (*a*,f(*a*)).

<u>Si la función admite derivada en todo punto</u> (o en toda una región), MAPLE puede computar la función derivada con el comando

D(f)(x)

donde f es la función previamente definida.

<u>Ejemplo</u>: Calcular la derivada de f(x)=$\sqrt{x^3}$

```
> f:=x->sqrt(x^3);
> D(f)(x);
```

Y podemos calcular valores concretos de dicha derivada, por ejemplo, la derivada en x=3, pues **D(f)(x)** es a su vez una función:

```
> D(f)(3);
```

Además, podemos calcular las derivadas sucesivas añadiendo la opción:

(D@@n)(f)(x)

donde n es el orden de derivación.

<u>Ejemplo:</u> Calcular las derivadas tercera y quinta de la función anterior:

```
> (D@@3)(f)(x);
```

```
> (D@@5)(f)(x);
```

También existe el comando **diff(f(x),x)**, pero da como resultado una variable, no una función, por lo que no lo usaremos en este curso.

Sin embargo, cuando la derivabilidad de una función no está clara, hay que echar mano de la definición:

Ejemplo: Estudiar la derivabilidad de la función f(x)= x^2 si $0 \le x$ y f(x)= 3x si $x < 0$.

Claramente esta función es derivable en todos los puntos distintos del 0 pues es polinómica, pero en el punto x=0 la derivabilidad es dudosa. Así que echamos mano de la definición con el límite. Como f(0+h) varía según sea h positivo o negativo, hay que hallar los límites laterales:

```
> limit(((0+h)^2-0^2)/h,h=0,right);
> limit(((3*h)-0^2)/h,h=0,left);
```

Por tanto, no es derivable en x=0.

Ejemplo: Estudiar la derivabilidad de f(x)=|x|.

En este caso sabemos que la definición del valor absoluto cambia en x=0, por lo que el comando D no nos funcionará. Así que tenemos que usar la definición con el cociente incremental:

```
> f:=x->abs(x);
> Limit((f(0+h)-f(0))/h,h=0)=limit((f(0+h)-f(0))/h,h=0);
```

Como no nos da el límite global, calculamos los límites laterales:

```
> Limit((f(0+h)-f(0))/h,h=0,left)=limit((f(0+h)-f(0))/h,h=0,lef
  t);
> Limit((f(0+h)-f(0))/h,h=0,right)=limit((f(0+h)-f(0))/h,h=0,ri
  ght);
```

Por tanto no existe el límite global en x=0, por lo que f no es derivable en 0. En los restantes puntos sabemos que si x > 0, f(x)=x y si x < 0, f(x)=-x, por lo que siempre es derivable en todo x no nulo.

Esto también se puede ver usando la representación gráfica de f(x):

```
> plot(f(x),x=-1..1);
```

en el que se ve claramente que en x=0 hay un punto anguloso por lo que f(x) no es derivable en dicho punto.

Recta tangente y normal a una curva

Si f(x) es derivable en un punto a, la **recta tangente a la curva** en dicho punto viene dada por la ecuación

$$y = \mathrm{f}(a) + \mathrm{D}(f)(a)(x - a)$$

Si f(x) es derivable en un punto a, la **recta normal a la curva** en dicho punto viene dada por la ecuación

$$y = f(a) - \frac{x - a}{D(f)(a)}$$

Ejemplo: Consideremos la función $f(x) = 3\,x^3\,(x^2 - 1)$. Hallar y representar gráficamente las rectas tangentes y normal a la curva en el punto x=-0.6.

```
> restart:
> f:=x->3*x^3*(x^2-1);
> tang:=x->f(-0.6)+ D(f)(-0.6)*(x+0.6);
> nor:=x->f(-0.6)- (1/D(f)(-0.6))*(x+0.6);
> plot([f(x),tang(x),nor(x)],x=-1..1,y=-1..1,color=[blue,gre
  en,red],title=`Rectas tangente y
  normal`,scaling=constrained);
> plot([f(x),tang(x),nor(x)],x=-0.8..-0.4,y=0.2..0.8,color=[
  blue,green,red],title=`Rectas tangente y
  normal`,scaling=constrained);
```

Monotonía y concavidad

Si f '(*a*) >0 entonces f(x) es **creciente** en un intervalo que contiene al punto *a*.

Si f '(*a*) <0 entonces f(x) es **decreciente** en un intervalo que contiene al punto *a*.

Si f '(*a*) =0 se dice que a es un **punto crítico** de f. El punto (*a*,f(*a*)) puede ser:

-Un **máximo** local de f (punto en que pasa de creciente a decreciente) si f ''(*a*) <0.
-Un mínimo local de f (punto en que pasa de decreciente a creciente) si f ''(*a*) >0.
-Un **punto de inflexión** de f si f ''(*a*) =0.

Si f ''(*a*) >0 entonces f(x) es **convexa** o cóncava hacia arriba en un intervalo que contiene el punto *a*.

Si f ''(*a*) <0 entonces f(x) es **cóncava** o cóncava hacia abajo en un intervalo que contiene el punto *a*.

Si f ''(*a*) =0 el punto (*a*,f(*a*)) es un **punto de inflexión** de f (punto donde cambia de convexa a cóncava o viceversa).

Ejemplo: Consideremos la misma función del ejemplo anterior $f(x) = 3\,x^3\,(x^2 - 1)$. Calcular sus intervalos de crecimiento y decrecimiento, sus puntos críticos y sus intervalos de concavidad y convexidad en el intervalo [-1,1]. Señalarlos en la gráfica de f(x).

```
> restart:
> f:=x->3*x^3*(x^2-1);
```

Primero calculamos sus puntos críticos calculando los puntos donde D(f)(x)=0. Se puede hacer mediante el método gráfico o mediante el método directo de MAPLE con el comando **solve**:

Método directo:

```
> sol1:=solve(D(f)(x)=0,x);
```

Para ver qué tipo de punto crítico son y estudiar los intervalos de crecimiento o decrecimiento, usamos la gráfica de la derivada:

```
> plot(D(f)(x),x=-1..1);
```

Claramente se ven los puntos críticos y el signo de f '(x), con lo que tenemos:

-en $[-1,-\frac{\sqrt{15}}{5})$ se tiene que f '(x) > 0, por lo que f es creciente.

-en $(-\frac{\sqrt{15}}{5},0)$ se tiene que f '(x) < 0, por lo que f es decreciente.

-en $(0,\frac{\sqrt{15}}{5})$ se tiene que f '(x) < 0, por lo que f es decreciente.

-en $(\frac{\sqrt{15}}{5},1)$ se tiene que f '(x) > 0, por lo que f es creciente.

Así podemos deducir que en $x=-\frac{\sqrt{15}}{5}$ hay un máximo y en $x=\frac{\sqrt{15}}{5}$ hay un mínimo, mientras que en x=0 vamos a tener un punto de inflexión. Para confirmarlo podemos aplicar la derivada segunda:

```
> (D@@2)(f)(-sqrt(15)/5);
```

Por tanto la derivada segunda es negativa en este punto crítico por lo que tenemos un máximo.

```
> (D@@2)(f)(sqrt(15)/5);
```

La derivada segunda es positiva en este punto crítico y tenemos un mínimo.

```
> (D@@2)(f)(0);
```

Aquí vemos que f ''(0) =0, por lo que en 0 tenemos un punto de inflexión como vimos en la gráfica.

NOTA: También podríamos usar la variable vectorial sol1 para poner (sqrt(15))/5 sin tenerlo que escribir todo:

```
> (D@@2)(f)(sol1[3]);
```

Para estudiar la convexidad estudiamos los puntos que anulan a la derivada segunda:

```
> sol2:=solve((D@@2)(f)(x)=0,x);
> evalf(sol2);
```

Estos son los tres puntos de inflexión que tiene f(x). Para ver los intervalos de convexidad y concavidad estudiamos su gráfica:

```
> plot((D@@2)(f)(x),x=-1..1);
```

Por tanto, tenemos:

-en $[-1,-\frac{\sqrt{30}}{10})$ y $(0,\frac{\sqrt{30}}{10})$ se tiene que f ''(x) <0 por lo que f(x) es cóncava (o cóncava hacia abajo).

-en $(-\frac{\sqrt{30}}{10},0)$ y $(\frac{\sqrt{30}}{10},1]$ se tiene f ''(x) >0 por lo que f(x) es convexa (o cóncava hacia arriba).

Si finalmente dibujamos f(x) podremos ver todas estas características en el dibujo, marcando los puntos críticos:

```
> with(plots):
```

Dibujamos los puntos críticos:

```
> g1:=plot([[sol1[2],f(sol1[2])],[sol1[3],f(sol1[3])],[sol1[
  4],f(sol1[4])],[sol2[2],f(sol2[2])],[sol2[3],f(sol2[3])]],
  style=point,symbol=diamond,color=black):
```

Dibujamos la gráfica de f(x):

```
> g2:=plot(f(x),x=-1..1,color=green):
```

Las mostramos las dos juntas:

```
> display(g1,g2);
```

Ejemplo: (Problema 14) Determinar la monotonía y convexidad de la función $f(x) = \frac{x^2+1}{x^2-4}$.

+ *Solución*

Máximos y mínimos

Una de las aplicaciones más importantes de la derivada es la de resolver problemas de optimización, esto es, localizar los valores extremos de una relación dada por una función de una variable.

En este tipo de problemas habitualmente se da una función de dos variables (pueden ser más) que hay que optimizar, a la vez que dichas variables deben verificar una condición conocida:

Ejemplo: (Problema 20) Un hoja de papel, de forma rectangular, debe contener 18 cm^2 de texto impreso. Los márgenes superior e inferior han de ser ambos de 2 cm; los laterales de 1 cm. ¿Cuáles deben ser las dimensiones de la página para que se use la menor cantidad de papel?

En este problema tenemos que minimizar el área de la hoja de papel, que es una función de dos variables, el ancho (a) y el largo (l) de la hoja, es decir:

```
> restart:
> A:=(a,l)->a*l;
```

La relación que conocemos que se tiene que cumplir es que dentro de la hoja tienen que haber 18 cm^2 cuando le quitemos 1+1=2 cm de margen al ancho y 2+2=4 cm de margen al largo, es decir, se tiene que verificar la ecuación algebraica:

```
> rel:=(a-2)*(l-4)=18;
```

Resolvemos esta ecuación en función de una de las variables. En MAPLE esto se puede hacer con el comando **solve** especificando la variable para la que queremos que la resuelva. En nuestro caso queremos despejar l en función de a:

```
> sol:=solve(rel,l);
```

Así podemos obtener el área en función de a solamente:

```
> A(a,sol);
```

Podemos dibujarla a ver si realmente va a tener un mínimo. Como a y l son distancias, sólo las dibujamos en el lado positivo:

```
> plot(A(a,sol),a=0..10);
```

Vemos que cuando a=2 hay una infinitud. Este comportamiento es lógico pues si a fuera 2 no cabría ningún texto en la hoja cuando le pusiéramos los 2 cm. de margen lateral. Así que lo dibujamos a partir de 2:

```
> plot(A(a,sol),a=2..10,y=0..100);
```

Así que gráficamente se ve que hay un mínimo para a en (4,6), donde la derivada se anula.

Para calcularlo tenemos que calcular la derivada de A(a,sol) respecto de a. Aquí tenemos un pequeño problema en MAPLE: para el programa A sigue siendo una función de dos variables, aunque asignemos l:=sol. Para transformarla en una función de una variable tenemos que introducir un nuevo comando:

unapply(expr,var)

que transforma la expresión **expr** en una función de la variable **var.**

Así transformamos la expresión de A(a,sol) en una función de una variable a:

```
> Aa:=unapply(A(a,sol),a);
```

Y ya sólo tenemos que hallar los puntos críticos de esta función calculando los ceros de su derivada:

```
> sold:=solve(D(Aa)(a)=0,a);
```

Así a=5 es el único punto crítico positivo. Por la gráfica ya sabemos que es un mínimo pero lo comprobamos viendo que la derivada segunda es positiva:

```
> (D@@2)(Aa)(5);
```

Para calcular el valor del largo l correspondiente hay que evaluar sol para a=5. Como sol no es una función sino una variable, hay que asignarle el valor a a:

```
> a:=5;
> l:=sol;
```

Y el valor mínimo del área es

```
> Aa(5);
```

<u>Ejemplo:</u> (Problema 22) Se van a usar 4 m. de alambre para formar un cuadrado y un círculo. ¿Qué cantidad de alambre debe usarse para el cuadrado y qué cantidad para el círculo a fin de abarcar la máxima área total.

Solución

Llamemos x al lado del cuadrado y r al radio del círculo. La longitud total de alambre del cuadrado más el del círculo será:

$$L = 4\,x + 2\,\pi\,r$$

y el problema nos dice que tiene que ser L=4. Por otro lado queremos optimizar (maximizar) el área total de las dos figuras que es

$$A = x^2 + \pi\,r^2$$

Así que vamos a definir esta función de dos variables y la ecuación algebraica que deben verificar dichas variables:

```
> restart:
> A:=(x,r)->x^2+Pi*r^2;
> rel:=4*x+2*Pi*r=4;
```

Resolvemos esta ecuación respecto de r:

```
> sol:=solve(rel,r);
```

Por tanto al sustituir esta relación en la función A(x,r) obtenemos una función que sólo depende de x:

```
> A(x,sol);
```

Para que MAPLE la considere una función de una variable x usamos

```
> Ax:=unapply(A(x,sol),x);
```

¿Esta función tiene un máximo para algún valor de x? Para ver si es así vamos a dibujarla. Para ello debemos considerar que si tenemos 4 m. de alambre y x es el lado del cuadrado, como mucho puede ser x=1 m. (en cuyo caso no tendríamos círculo). Así que los únicos valores que nos sirven son para x en [0,1]:

```
> plot(Ax(x),x=0..1);
```

Por tanto, el único punto con derivada nula en [0,1] en este caso es un mínimo, así que no nos vale. Pero, ¿Ax alcanza un máximo en [0,1]? Sí: en x=0, aunque no anule la derivada, el punto (0,Ax(0)) es un máximo en [0,1]. Por tanto, la solución es

```
> x:=0;
> r:=sol;
```

Es decir, que para alcanzar el máximo de área no debemos hacer un cuadrado, sino sólo un círculo de radio $r=\frac{2}{\pi}$ con lo que tendríamos un área máxima

```
> A(x,r);
```

Ejemplo: Un depósito abierto de chapa con fondo cuadrado debe tener capacidad para V litros. ¿Qué dimensiones debe tener dicho depósito para que en su fabricación se necesite la menor cantidad de chapa?

+ ***Solución***

Polinomio de Taylor

Dada una función f(x) que admite derivadas de cualquier orden en un punto a podemos construir un polinomio p(x) de grado n que se aproxima a dicha función en un entorno del punto a. Este polinomio se denomina **polinomio de Taylor de grado n** y viene dado por

$$\mathrm{p}(x)=f(a)+\frac{1}{1!}f'(a)\,(x-a)+\frac{1}{2!}f''(a)\,(x-a)^2+\frac{1}{3!}f'''(a)\,(x-a)^3+\ldots+\frac{1}{n!}f^{n}(\mathrm{a})$$

Si $a=0$, dicho polinomio se suele llamar **polinomio de MacLaurin** de grado n.

Además, podemos obtener una medida del error de aproximación que se comete mediante el llamado **resto de Lagrange**:

$$\mathrm{f}(x)=\mathrm{p}(x)+\mathrm{R}(x)$$

donde

$$R(x) = \frac{1}{(n+1)!} f^{(n+1)}(\mathbf{t})\ (x-a)^{(n+1)}$$

para cierto *t* en *(a,x)* ó *(x,a)*.

Ejemplo: Determinar diversos polinomios de Taylor en a=0 y a=2 de la función $f(x) = \mathrm{sen}(x) - x\cos(x)$.

```
> restart:
> f:=x->sin(x)-x*cos(x);
```

Definimos el polinomio de Taylor de f(x) en a =0 (o polinomio de MacLaurin) y en **a**=2:

```
> p0:=(x,n)->f(0)+sum((D@@k)(f)(0)*x^k/k!,k=1..n);
> p2:=(x,n)->f(2)+sum((D@@k)(f)(2)*(x-2)^k/k!,k=1..n);
> p0(x,3);
> p0(x,8);
> p2(x,5);
> p2(x,7);
```

Representamos gráficamente varios polinomios de MacLaurin junto con la función f(x):

```
> with(plots):
> g1:=plot(f(x),x=-2..2,color=blue):
> g2:=plot(p0(x,3),x=-2..2,color=red):
> g3:=plot(p0(x,5),x=-2..2,color=black):
> g4:=plot(p0(x,9),x=-2..2,color=green):
> display(g1,g2,g3,g4);
```

A medida que nos alejamos de x=0 la aproximación es peor:

```
> g1:=plot(f(x),x=-5..5,color=blue):
> g2:=plot(p0(x,3),x=-5..5,color=red):
> g3:=plot(p0(x,5),x=-5..5,color=black):
> g4:=plot(p0(x,9),x=-5..5,color=green):
> display(g1,g2,g3,g4);
```

Ejemplo: Con la misma función del ejemplo anterior, estimar una cota del error cometido por el polinomio de MacLaurin de grado 5 en el punto x=1. Comparar con el error que da MAPLE.

```
> p0(x,5);
```

Error de Lagrange del polinomio de MacLaurin de grado 5:

```
> r:=(x,t)->(D@@6)(f)(t)*x^6/6!;
```

Como en este caso a=0 y x=1, t puede ser cualquier número en (0,1). Por tanto, sólo podremos afirmar que el error será como mucho el máximo de r(1,t) (en valor absoluto) cuando t varía en (0,1). MAPLE puede calcularnos dicho máximo con el comano

maximize (función,{variable},{rango})

En este comando **función** es la función que queremos maximizar, **variable** es el nombre de la variable (o variables) respecto la que queremos maximizar dicha función y **rango** es el intervalo donde varía dicha variable. Es importante no olvidar las llaves al poner la variable y el rango.

En nuestro caso, queremos calcular el máximo de abs(r(1,t)) para t variando en (0,1). Así la orden queda:

```
> errorlag:=evalf(maximize(r(1,t),{t},{t=0..1}));
```

Por tanto el resto de Lagrange nos dice que el error en valor absoluto lo peor que puede ser es 0 .006593968374.

Para comprobar si esta estimación es buena, calculemos el error que da MAPLE:

```
> error:=evalf(abs(f(1)-p0(1,5)));
```

Así que es cierto que el error está acotado por 0.006593968374, por lo que la estimación del resto de Lagrange es apropiada.

Ejemplo: (Problema 38) Estimar el error cometido al aproximar el número *e* mediante el polinomio de MacLaurin de grado 5.

+ *Solución*

Ejemplo: (Problema 37) Hallar el polinomio de Taylor centrado en a=1 que debe usarse para aproximar ln(1.2) con un error menor que 0.001.

− *Solución*

En este caso, f(x)=ln(x), a=1, x=1.2. Así el polinomio de Taylor en general es

```
> restart:
> f:=x->ln(x);
> p:=(x,n)->f(1)+ sum((D@@k)(f)(1)*(x-1)^k/k!,k=1..n);
```

Y el resto de Lagrange del polinomio de grado n es (aquí también dependerá de n):

```
> r:=(x,t,n)-> (D@@(n+1))(f)(t)*(x-1)^(n+1)/(n+1)!;
```

Como en nuestro caso x=1.2

```
> r(1.2,t,n);
```

Como en nuestro caso x=1.2, vamos a ir calculando los sucesivos restos de los polinomios para n=1,2,3,... viendo si el error que proporcionan es menor que 0.001 o no. Como dichos restos dependen de t que varía en (1,1.2) sin que sepamos quién es, vamos obteniendo los máximos de los restos en valor absoluto para los n sucesivos:

```
> maximize(abs(r(1.2,t,1)),{t},{t=1..1.2});
```

Como el error no es menor que 0.001 lo intentamos con el de grado 2:

```
> maximize(abs(r(1.2,t,2)),{t},{t=1..1.2});
```

Sigue siendo mayor que 0.001, por lo que lo intentamos con el de grado 3:

```
> maximize(abs(r(1.2,t,3)),{t},{t=1..1.2});
```

Este valor sí que es menor que 0.001, por lo que el polinomio de Taylor de grado 3 ya nos da un error menor que 0.001. Comprobémoslo:

```
> evalf(abs(f(1.2)-p(1.2,3)));
```

− *Nota*

En MAPLE hay una forma mejor de ir calculando los sucesivos máximos del ejemplo anterior sin tener que ir repitiendo una a una la misma orden para valores de n distintos: los bucles condicionales

while *condición* do..od

Así se crea un pequeño algoritmo que se escribe como:

```
> n:=0;
  errorlag:=1;
  while errorlag >= 0.001 do
     n:=n+1;
     errorlag:=maximize(abs(r(1.2,t,n)),{t},{t=1..1.2});
  od;
```